Structural Model for Textile

dr.ir. P.C.J. Hoogenboom
June 2007

Delft University of Technology
The Netherlands

Preface

In the M.Sc. project of Mr. P. van Asselt bi-axial loading tests have been performed on textiles for tent structures. The stresses were plotted as function of the strains. This showed that the data did no resemble linear elastic behaviour but was quite smooth. Nonetheless, it we were not successful in fitting curved surfaces through the data. Therefore, we left the phenomenological approach and developed the model in this report. It took a few days before we had all equations right and a found a successful solution algorithm. The intention is to match this model to the experimental data, implement it in ANSYS (usermat) and analyse the structural behaviour of tent structures.

Equations

d	diameter of the undeformed wires
s	spacing of the undeformed wires
α	linear temperature extension coefficient
ΔT	temperature increase

We assume the same properties in the x direction as the y direction.



Kinematic equations

$$\varepsilon_{xx} = \frac{s_x - s}{s}$$
$$\varepsilon_{yy} = \frac{s_y - s}{s}$$
$$a^2 = s^2 + d^2$$
$$a_x^2 = s_x^2 + (d - w)^2$$
$$a_y^2 = s_y^2 + (d + w)^2$$

Constitutive equations

$$T_x = EA(\frac{a_x - a}{a} - \alpha \Delta T)$$
$$T_y = EA(\frac{a_y - a}{a} - \alpha \Delta T)$$
$$A = \frac{1}{4}\pi d^2$$

Equilibrium equations

$$\frac{T_x}{N} = \frac{a_x}{d - w}$$
$$\frac{T_y}{N} = \frac{a_y}{d + w}$$
$$\frac{N_x}{T_x} = \frac{s_x}{a_x}$$
$$\frac{N_y}{T_y} = \frac{s_y}{a_y}$$
$$n_{xx} = \frac{N_x}{s}$$
$$n_{yy} = \frac{N_y}{s}$$

Solution

These nonlinear equations can be solved with a small computer program. The program uses iterations until the unbalance *R* in the force *N* is sufficiently small. In most situations a few iterations are required and never more than 20.
In addition, the program includes that the wires can only carry tension forces.

```
> # stresses in membrane material
> # -- input ---------------------------------------------------------
> E:=1000.:           # [N/mm2] Young's modulus of the wires plus matrix
> G:=500.:            # [N/mm2] shear modulus of the matrix
> d:=0.5:             # [mm]    diameter of the wires
> s:=1.5:             # [mm]    spacing of the wires (0<2d<s)
> epsilonxx:=0.030:   # [-]     strain in the x-direction
> epsilonyy:=0.002:   # [-]     strain in the y-direction
> gammaxy:=0.001:     # [-]     shear strain in the xy-direction
> epsilonxxp:=0.0:    # [-]     strain due to prestress in x-direction
> epsilonyyp:=0.0:    # [-]     strain due to prestress in y-direction
> alpha:=0.0:         # [-]     linear temperature extension coefficient
> dT:=0.0:            # [C]     temperature increase
> # -- computation ---------------------------------------------------
> a:=sqrt(s^2+d^2):
> A:=1/4*3.1415*d^2:
> sx:=s*(1+epsilonxx+epsilonxxp):
> sy:=s*(1+epsilonyy+epsilonyyp):
> w:=0: R:=100:
> while abs(R)>0.000001*a*d do
     ax:=sqrt(sx^2+(d-w)^2):
     ay:=sqrt(sy^2+(d+w)^2):
     Tx:=(ax/a-1-alpha*dT): if (Tx<0) then Tx:=0: end if:
     Ty:=(ay/a-1-alpha*dT): if (Ty<0) then Ty:=0: end if:
     R:=Tx*ay*(d-w)-Ty*ax*(d+w):
     w:=w+R/a:
  end do:
> Tx:=E*A*Tx:
> Ty:=E*A*Ty:
> Nx:=sx/ax*Tx:
> Ny:=sy/ay*Ty:
> nxx:=Nx/s:
  nyy:=Ny/s:
  nxy:=G*d*gammaxy:
> # -- output ----------------------------------------------------------
> nxx;
  nyy;
  nxy;
```

It can be shown that the normal forces are linear in Young’s modulus *E* and the wire diameter *d*.

$$n_{xx} = Ed\, f(\frac{s}{d}, \varepsilon_{xx}, \varepsilon_{yy})$$

$$n_{yy} = Ed\, f(\frac{s}{d}, \varepsilon_{yy}, \varepsilon_{xx})$$

The following figure shows the function *f* as a function of ε_{xx} and ε_{yy} for *s* = 3*d*.



Just two experimental measurements are required to determine *Ed* and $\frac{s}{d}$. For example,

$n_{xx} = n_{yy} = n$ and $n_{xx} = n$, $n_{yy} = 0$.

Prestress

In the simulation of tent behaviour the material is first prestressed. To this end the strains that results from stresses (prestress) needs to be computed. This is the inverse problem of the one solved above. The program below shows an algorithm for this.

```
> # strains in membrane material
> # -- input ---------------------------------------------------------
> E:=1000.:           # [N/mm2] Young's modulus of the wires plus matrix
> d:=0.5:             # [mm]    diameter of the wires
> s:=1.5:             # [mm]    spacing of the wires (0<2d<s)
> nxx:=6.30:          # [N/mm]  force in the x-direction
> nyy:=8.20:          # [N/mm]  force in the y-direction
> # -- computation ---------------------------------------------------
```

```
> a:=sqrt(s^2+d^2):
> A:=1/4*3.1415*d^2:
> Nx:=nxx*s: if (Nx<0) then Nx:=0: end if:
> Ny:=nyy*s: if (Ny<0) then Ny:=0: end if:
> w:=d*(Nx-Ny)/(Nx+Ny+0.00000001):
> Tx:=Nx:
> Ty:=Ny:
> for i from 1 to 4 do
    ax:=a*(1+Tx/(E*A)):
    ay:=a*(1+Ty/(E*A)):
    sx:=sqrt(ax^2-(d-w)^2):
    sy:=sqrt(ay^2-(d+w)^2):
    w:=d*(Nx*sy-Ny*sx)/(Nx*sy+Ny*sx+0.00000001):
    Tx:=ax/sx*Nx:
    Ty:=ay/sy*Ny:
  end do:
> epsilonxxp:=sx/s-1:
> epsilonyyp:=sy/s-1:
> # -- output ----------------------------------------------------------
> epsilonxxp;
  epsilonyyp;
```

0.043979377

0.082899771